
Acest modul al cursului ne ajută să înţelegem principalele avantaje ale funcţiilor din grupurile 
COUNT și SUM. Cursul conţine o serie de exemple de la cele mai simple la cele avansate.  

Introducere 

Asemănător cu formulele și problemele matematice, formulele Excel au la bază propria 

sintaxă și logică. Odată ce înţelegi sintaxa unei formule, poţi folosi aceste formule pentru 

a salva timp prin automatizarea anumitor calcule și activităţi.  

Unele dintre cele mai folosite operaţii în Excel sunt cele legate de însumarea și 

numărarea celulelor. 

În cadrul acestui modul, vom învăţa cum să folosim principalele funcţii pentru 

însumarea și numărarea în cadrul unei zone de celule sau în funcţie de anumite condiţii. 

Metode simple de adunare și numărare a 

celulelor 

Adunarea valorilor într-o celulă folosind o formulă simplă 

Dacă ai nevoie de un rezultat rapid, programul Excel poate fi folosit ca un mini 

calculator. Puteţi face acest lucru prin folosirea operatorului aritmetic +.  

Pentru adunarea a două sau a mai multor valori sau celule, este nevoie să creăm o 

formulă. 

Ce trebuie să știm atunci când realizăm o formulă în Excel: 

• Toate formulele încep cu semnul egal = 

• Semnul egal se va scrie mereu în celula în care se dorește realizat calculul 

• Formula se va calcula după apăsarea tastei ENTER 

Exemplu 1 -  formulă aritmetică 

În cadrul acestui exemplu, vom folosi o formulă aritmetică simplă pentru a aduna 

vânzările din 2012 și 2013 din ţara US, regiunea South. 

 

Adunarea valorilor de pe un rând sau o coloană folosind AUTOSUM  

De foarte multe ori, este nevoie să calculăm totalul valorilor existente pe o coloană sau 

pe un rând. Pentru a realiza acest lucru rapid putem folosi funcţia AutoSum. Când 

funcţia este accesată, aceasta va sugera zona de celule din care vor fi adunate valorile, 

dacă sugestia nu este corectă puteţi modifica formula autocompletată de AutoSum. 

Pentru calcularea valorii totale a vânzărilor din tabelul exemplu folosit pentru funcţiile 

de tip SUM, este nevoie să urmăm următorii pași: 

1. Selectăm o celulă goală la dreapta sau sub datele pe care dorim să le adunăm. 

Nu este obligatoriu ca celula selectată să fie prima celulă goală, însă zona de 

celule dintre valori și cea în care vom adăuga AutoSum trebuie să fie goală. În 

exemplul nostru, vom selecta celula F31. 

2. Apăsăm butonul AutoSum. În variantele Excel 2007/2010/2013, acest buton se 

află în tab-ul Home din bara de meniu.  

 

3. Excel va adăuga în mod automat funcţia SUM cu zona de celule numerice 

selectate. În exemplul nostru, programul va completa toate celule numerice pe 

care le întâlnește până la prima non numerică F7:F30. 

Numărarea celulelor numerice de pe un rând sau o coloană 

folosind AUTOCOUNT 

O funcţie asemănătoare AutoSum-ului este AutoCount care ne ajută să numărăm rapid 

celulele numerice de pe o coloană sau un rând. 

În exemplul nostru, vom calcula numărul de valori numerice aflate pe coloana Total din 

tabelul folosit pentru funcţiile de tip COUNT. Pentru acest lucru vom urma o serie de 

pași astfel: 

1. Selectăm o celulă goală la dreapta sau sub datele pe care dorim să le numărăm. 

Noi vom selecta celula G21. 

2. Apăsăm săgeata aflată lângă butonul AutoSum. Din lista care se deschide vom 

alege Count Numbers. 

 

3. Excel va adăuga în mod automat funcţia COUNT cu zona de celule adiacente pe 

care le sugerează. Dacă zona de celule nu este corectă, va fi nevoie să se modifice 

manual în cadrul formulei. 

În cazul nostru, va număra toate celulele numerice din zona G7:G20 și va returna 

valoarea 14 pentru că toate valorile sunt numerice. 

Elemente de bază ale funcţiilor de tip SUM 

Sintaxa funcţiei SUM () 

Funcţia Excel SUM se folosește pentru adunarea unui set de valori date.  

SUM(number1,[number2],...) 

• Number – reprezintă tipul unic de argumente al funcţiei SUM. Acesta poate să fie 

reprezentat de valori introduse manual de utilizator sau de referinţe către alte 

celule. 

În versiunile Excel după 2007, se pot introduce maxim 255 de argumente în funcţia SUM. 

Iar în versiunile anterioare se pot introduce maxim 30 de argumente. Cu toate acestea, 

fiecare argument poate să reprezinte o zonă de celule sau un array de valori. 

Tipurile de date care pot fi folosite în cadrul funcţiei SUM 

De cele mai multe ori, funcţia SUM este folosită pentru a aduna date de tip numeric sau 

de tip dată.  De fapt, valorile tip dată sunt salvate ca numere. Programul Excel alocă 

pentru o dată numărul de zile care au trecut de la 01.01.1900. Numărul 1 este 

01.01.1900, 2 este 02.01.1900 și așa mai departe. 

În tabelul de mai jos, putem vedea care tipuri de date sunt adunate, ignorate sau care 

rezultă în erori: 

Tip de dată   
Valori din cadrul unei zone de 

celule 

Valori introduse manual de 

utilizator 

Numere Vor fi adunate Vor fi adunate 
Date Vor fi adunate Vor fi adunate 

Valori logice Vor fi adunate 

Vor fi adunate (True=1; 
False=0) 
Exemplu: =SUM(TRUE,TRUE) 
Rezultat 2 

Text care 

reprezintă numere 

sau date 

Vor fi adunate 
Vor fi adunate.  
Exemplu: =SUM("8", "9" ) 
Rezultat 17 

Alte valori tip text Vor fi ignorate Va rezulta eroarea #VALUE! 

Exemplu 2 -  funcţia SUM 

În cadrul acestui exemplu, vom folosi tabelul de date din sheet-ul SUM și dorim să aflăm 

valoarea totală a vânzărilor din anul 2012. 

=SUM(F7:F10,F15:F18,F23:F26) 

 

Sintaxa funcţiei SUMIF () 

Atunci când dorim să adunăm anumite valori în funcţie de o condiţie, avem nevoie de 

funcţia SUMIF. În continuare vom înţelege sintaxa acestei funcţii și vom vedea cum 

putem să o folosim în activităţile noastre zilnice. 

SUMIF(range, criteria, [sum_range]) 

Sintaxa funcţiei SUMIF conţine 3 argumente. Dintre care primele 2 sunt obligatorii, iar 

ultimul este opţional. 

• Range – reprezintă zona de celule care va fi evaluată în funcţie de condiţia notată 

în argumentul Criteria. 

• Criteria – reprezintă condiţia care definește celulele care vor fi adunate. Condiţia 

dorită poate să fie de tip numeric, text, de tip dată, valoare logică sau 

reprezentând altă celulă. De exemplu, putem introduce condiţia astfel: “5”, 

“Canada”, “1/1/2012”, “<5”, A1 etc. 

• Sum_range – reprezintă celulele care vor fi adunate dacă condiţia setată mai sus 

este îndeplinită de celulele selectate la argumentul Range. Dacă acest argument 

este omis, celulele din Range se folosesc atât pentru evaluarea condiţiei cât și 

pentru adunare. 

Pentru a ilustra mai bine sintaxa funcţiei SUMIF, haideţi să considerăm următorul 

exemplu. 

Exemplu 3 - funcţia SUMIF 

Dorim să aflăm valoarea totală a vânzărilor dintr-o anumită ţară. Argumentele funcţiei 

SUMIF sunt următoarele: 

• Range –  $D$7:$D$30 

• Criteria – putem scrie numele ţării direct în formulă “Canada” sau putem 

introduce condiţia într-o altă celulă: $J$13 

• Sum_range – $F$7:$F$30 

=SUMIF($D$7:$D$30,$J$13,$F$7:$F$30) 

Exemple 4-8  Mai mic sau mai mare decât o valoare dată 

În cadrul funcţiei SUMIF, argumentul de tip condiţie poate fi folosit împreună cu 

operatorii de comparare: 

Condiţia Operator Formula exemplu Descriere 

Mai mare 

decât o 

valoare 

> =SUMIF($F$7:$F$30,">1000") 
Adună toate valorile de pe 
coloana Valoare vanzari ($) 

care sunt mai mari ca 1000. 

Mai mic decât 

o valoare 
< =SUMIF($F$7:$F$30,"<500") 

Adună toate valorile de pe 
coloana Valoare vanzari ($) 

care sunt mai mici de 500. 

Diferit de o 

valoare 
<> 

=SUMIF($D$7:$D$30,"<>Canada", 
$F$7:$F$30) 

Adună toate vânzările 
pentru ţările diferite de 
Canada. 

Mai mare sau 

egal decât o 

valoare 

>= =SUMIF($F$7:$F$30,">=1000") 

Adună toate valorile de pe 
coloana Valoare vanzari ($) 

care sunt mai mari sau 
egale ca 1000. 

Mai mic sau 

egal decât o 

valoare 

<= =SUMIF($F$7:$F$30,"<=750") 

Adună toate valorile de pe 
coloana Valoare vanzari ($) 

care sunt mai mici sau egale 
de 750. 

Sintaxa funcţiei SUMIFS () 

Funcţia SUMIFS reprezintă varianta îmbunătăţită a funcţiei SUMIF și o putem folosi 

pentru a aduna valori pe baza de condiţii multiple. Această funcţie a fost introdusă o 

dată cu versiunea din 2007 a programului. 

 

=SUMIFS(sum_range, criteria_range1, criteria1, [criteria_range2, 

criteria2], …) 

• Sum_range – reprezintă celulele care vor fi adunate dacă condiţiile setate sunt 

îndeplinite. 

• Criteria_range1 – reprezintă zona de celule care vor fi evaluate în funcţie de 

condiţia 1. 

• Criteria1 – reprezintă condiţia 1 care definește celulele care vor fi adunate. 

• Criteria_range2, criteria2, …  - sunt argumente opţionale și reprezintă alte 

condiţii și celule de comparare suplimentare. Pot fi introduse până la 127 perechi 

de Criteria_range și Criteria.  

Exemplu 9 – Funcţia SUMIFS 

Pe baza tabelului de date din sheet-ul SUM, vom calcula valoarea totală a vânzărilor 

dintr-un an dat și dintr-o anumită ţară. Argumentele funcţiei SUMIFS sunt următoarele: 

• Sum_range –  $F$7:$F$30 

• Criteria_range1 - $C$7:$C$30 

• Criteria1 – putem scrie anul direct în formulă 2012 sau putem introduce condiţia 

într-o altă celulă: $N$7 

• Criteria_range2 - $D$7:$D$30 

• Criteria2 – putem scrie numele ţării direct în formulă “US” sau putem introduce 

condiţia într-o altă celulă: $P$7 

=SUMIFS($F$7:$F$30,$C$7:$C$30,$N$7,$D$7:$D$30,$P$7)

 

Exemplu 10 – Folosirea caracterelor de tip Wildcard 

La fel ca funcţiile SUMIF, COUNTIF, și în cadrul funcţiei SUMIFS condiţiile de tip text se 

pot folosi împreună cu caractere de tip Wildcard. 

Tip Wildcard Descriere 

? 
Exemplu: ”u?u” 
Semnul ? înlocuiește o singură literă dintr-un text unu. 

* 

Exemplu: ”d*a” 
Semnul * înlocuiește orice număr de litere dintr-un text (data, 

duminica, direcţia). 

De exemplu, pe tabelul de date din sheet-ul SUM vom calcula valoarea totală a 

vânzărilor dintr-o ţară dată pentru toate regiunile care se termină în th. Pentru a realiza 

acest lucru vom folosi caracterul *. 

=SUMIFS($F$7:$F$30,$D$7:$D$30,$O$13,$E$7:$E$30,"*th") 

Utilizări avansate ale funcţiilor din grupul SUM 

Adunarea valorilor în funcţie de un interval dat de timp 

Deseori avem nevoie să calculăm totalul unor valori realizat între două date. Există două 

variante prin care putem să aflăm această sumă: una folosind funcţia SUMIF (pentru 

Excel 2003 sau o versiune anterioară) sau, mult mai ușor, folosind funcţia SUMIFS 

(pentru Excel 2007 sau o versiune următoare) 

Exemplu 11 – Adunarea valorilor în funcţie de un interval de timp folosind SUMIF 

Pentru a realiza adunarea valorilor în funcţie de o perioadă dată de timp vom folosi 

două funcţii SUMIF. Pentru acest exemplu, vom folosi tabelul de date 1 din sheet-ul 

SUM Avansat și dorim să aflăm valoarea totală a cheltuielilor realizate între 01.01.2015 

și 05.01.2015. 

=SUMIF($C$7:$C$27,">="&$J$7,$G$7:$G$27)-

SUMIF($C$7:$C$27,">"&$L$7,$G$7:$G$27) 

La prima vedere formula poate părea complicată, însă dacă înţelegem mecanismul din 

spatele ei, va fi foarte simplu de aplicat pe viitor. Prima funcţie SUMIF adună toate celule 

de pe coloana Valoare al căror corespondent pe coloana Dată este mai mare decât 

01.01.2015.  

Din rezultatul primit, scădem suma valorilor realizate după data de sfârșit (05.01.2015). 

Pentru a calcula această sumă folosim a doua funcţie SUMIF. 

Exemplu 12 – Adunarea valorilor în funcţie de un interval de timp folosind SUMIFS 

Odată cu introducerea funcţiei SUMIFS în versiunea din 2007 a programului Excel, 

artificiul realizat cu folosirea a două funcţii SUMIF nu mai este necesar. Formula de mai 

sus se comprimă într-o simplă formulă SUMIFS cu două condiţii: 

=SUMIFS($G$7:$G$27,$C$7:$C$27,">=" & $J$11,$C$7:$C$27,"<=" & $L$11) 

 
 

Adunarea valorilor de pe mai multe coloane (exemplu 13) 

Această modalitate de calcul se pretează cel mai mult în cazul tabelelor care 

centralizează informaţiile pe luni, ani sau alte categorii. Pentru acest exemplu, vom 

folosi tabelul 2 din sheet-ul SUM Avansat care centralizează vânzările pe ţară, regiune și 

an sub forma următoare: 

Să presupunem că dorim să aflăm valoarea vânzărilor realizate în Canada în anii 2012 și 

2013. Este posibil să fim tentaţi să folosim o singură funcţie SUMIF, însă atât aceasta cât 

și SUMIFS nu pot aduna valori de pe mai multe coloane. 

O primă variantă ar fi să adunăm mai multe funcţii SUMIF: 

=SUMIF($C$32:$C$43,$J$32,$E$32:$E$43)+ 

SUMIF($C$32:$C$43,$J$32,$F$32:$F$43) 

O altă variantă mult mai elegantă presupune folosirea variantei array a funcţiei SUM. Nu 

uitaţi sa apăsaţi CTRL + SHIFT + ENTER la final. 

{=SUM(($E$32:$E$43+$F$32:$F$43)*(--($C$32:$C$43=$J$32)))}  

 
Ambele variante vor avea același rezultat, și anume 9130. 

Adunarea valorilor de pe mai multe coloane dinamice (exemplu 14) 

În cadrul exemplului anterior, am văzut cum putem aduna valorile de pe mai multe 

coloane în funcţie de o condiţie dată. 

Însă există situaţii în care este nevoie ca un alt utilizator să aleagă coloanele de pe care 

adunăm. Astfel, formula de mai sus va fi una dinamică. Pentru acest lucru, ne vom folosi 

și de combinaţia INDEX-MATCH de care am vorbit în cadrul modulului 2 al acestui curs. 

{=SUM((INDEX($C$32:$G$43,,MATCH($J$37,$C$31:$G$31,0))+ 

INDEX(C32:G43,,MATCH($L$37,$C$31:$G$31)))*(--($C$32:$C$43=$J$36)))} 

În exemplul de mai sus, am ales manual coloanele de pe care dorim adunarea valorilor. 

În situaţia de faţă, ne vom folosi de combinaţia de funcţii INDEX-MATCH pentru a 

returna în mod automat o zonă de celule în funcţie de anii notaţi în celulele J37 și L37. 

Mai departe o să vedem modalitatea în care Excel va calcula această formulă: 

1. Prima funcţie calculată este MATCH($J$37,$C$31:$G$31,0) – rezultat 4 care oferă 

funcţiei INDEX numărul coloanei de pe care extrage informaţiile. 

2. Al doilea pas este calcularea primei funcţii INDEX care va avea ca rezultat zona de 

celule pentru primul an din condiţie INDEX($C$32:$G$43,,4) – 

{360;1410;220;540;750;1230;1470;1350;1460;740;750;810}. Trebuie reţinut faptul 

că la primul argument din INDEX selectăm tot tabelul de date și că la argumentul 

row_num nu am notat nimic pentru că dorim să extragem toată zona de celule.  

3. Urmează calcularea celei de-a doua funcţii MATCH($L$37,$C$31:$G$31)) – 

rezultat 5 care oferă funcţiei INDEX numărul coloanei de pe care extrage 

informaţiile pentru al doilea an din condiţie. 

4. Apoi se va calcula al doilea INDEX care va avea ca rezultat zona de celule pentru 

al doilea an din condiţie INDEX($C$32:$G$43,,5) – 

{460;1200;300;650;900;1120;1780;1560;1320;860;740;960}. Programul Excel, va 

păstra în memorie cele două arii de valori. 

5. În continuare va fi analizată partea (--($C$32:$C$43=$J$36)) prin care se compară 

fiecare celulă de pe coloana Ţară cu celula în care este notată condiţia pentru 

ţară. Rezultatul acestei comparaţii va fi un șir de True și False adică 1 și 0. 

{0;0;0;0;1;1;1;1;0;0;0;0}. 

6. De abia acum intră în acţiune funcţia SUM. Din fiecare arie de valori păstrată se 

vor alege doar cele de pe aceeași poziţie cu cele care au 1. 

 

Adunarea valorilor pe bază de condiţii multiple folosind funcţia 

SUM (exemplu 15) 

Această sarcină a devenit foarte ușoară odată  cu introducerea funcţiei SUMIFS. Însă 

aceste calcule sunt necesare și pentru cei care încă folosesc o versiune Excel 2003 sau 

anterioară. 

În continuare, vom vedea cum putem folosi funcţia SUM pentru adunarea valorilor pe 

bază de condiţii multiple. Pentru acest exemplu, vom folosi tabelul 1 de date din sheet-ul 

SUM Avansat și dorim să calculăm valoarea totală a cheltuielilor realizate de Persoana 

1 pe categoria Mancare si alte necesitati. 

• Condiţia 1: Coloana Titular $B$7:$B$27 = “Persoana 1” ($J$15) 

• Condiţia 2: Coloana Categorie $C$7:$C$27 = “Mancare si alte necesitati” ($L$15) 

Pentru aplicarea mai multor condiţii în interiorul funcţiei SUM, vom scrie o formulă care 

să multiplice cele două condiţii fiecare între un set de paranteze. La sfârșit, vom adăuga 

sub aceeași formă, zona de celule din care dorim să facem adunarea. Formula care 

rezultă este una de tip Array și este nevoie să folosim la final CTRL + SHIFT + ENTER. 

{=SUM(($D$7:$D$27=$J$15)*($E$7:$E$27=$L$15)*($G$7:$G$27))} 

 

Adunarea primelor n valori dintr-o listă 

Atunci când analizăm diverse tabele de date, există momente când dorim să știm 

valoarea celor mai mari n din listă. De exemplu, dacă am avea o listă cu cifrele de afaceri 

ale tuturor clienţilor dintr-un an, putem calcula care este valoarea realizată de cei mai 

mari 5 clienţi.  

O variantă ar fi să sortăm lista în mod descrescător și să facem o sumă simplă pe 

primele 5 celule. Însă nu este o soluţie prea elegantă. În continuare vom trece prin 3 

rezolvări care folosesc funcţiile SUM și LARGE. 

  



 FUNCŢIILE EXCEL DIN GRUPURILE COUNT ȘI SUM 

 
 19 | P a g e  

Exemplu 16 – Adunarea primelor 3 valori din listă (manual) 

Dacă problema noastră necesită de fiecare dată calcularea primelor 3 valori din tabel, 

putem introduce acest număr manual în cadrul funcţiei LARGE. . Pentru acest exemplu, 

vom folosi tabelul 2 din sheet-ul SUM Avansat pentru a calcula totalul primelor 3 

vânzări din anul 2014. 

{=SUM(LARGE($G$32:$G$43,{1,2,3}))} 

Folosim funcţia LARGE pentru a extrage primele 3 valori din tabel. Deoarece aceasta va 

returna o arie de valori, formula este una de tip Array și este nevoie să folosim  la final 

CTRL + SHIFT + ENTER. 

 

Prin înlocuirea funcţiei LARGE cu SMALL putem afla valoarea ultimelor 3 valori din listă.  

Exemplu 17 – Adunarea primelor 10 valori din listă (static) 

Dacă dorim să calculăm valorile pentru primele n valori din listă, unde n este un număr 

mai mare de 5, în cadrul funcţiei LARGE vom înlocui partea {1,2,3….,10} cu funcţia ROW.  

Pe baza aceluiași set de date, calculăm suma primelor 10 vânzări din anul 2014 folosind 

formula următoare: 

{=SUM(LARGE($G$32:$G$43,ROW(N1:N10)))} 

Funcţia ROW oferă ca rezultat numărul de rând al celulei selectate. Dacă este selectată o 

zonă de celule care începe pe un rând și se termină pe altul, rezultatul va include toate 

numerele de rând prezente în selecţie. 

În cadrul formulei noastre, am specificat zona de celule N1:N10 pentru a primi ca 

rezultat toată aria de valori {1,2,3….,10}. În loc de N1:N10 puteţi selecta orice zonă de 

celule, important este ca aceasta să înceapă de pe rândul 1 și să se termine la rândul 

până unde doriţi să fie calculată suma.  

Formula este una de tip Array și este nevoie să folosim  la final CTRL + SHIFT + ENTER. 

 

Prin înlocuirea funcţiei LARGE cu SMALL putem afla valoarea ultimelor 10 valori din 

listă.  

Exemplu 18 – Adunarea primelor n valori din listă (unde n este dinamic) 

Cea mai elegantă soluţie pentru a realiza acest calcul este atunci când utilizatorul 

fișierului are posibilitatea de a completa într-o celulă valoarea pentru n.  

Vom realiza acest lucru folosind combinaţia de funcţii de la exemplul anterior la care 

includem și funcţia INDIRECT. Această funcţie se utilizează pentru crearea unei referinţe 

către o celulă sau o zonă de celule. În exemplul nostru, ne vom folosi de ea pentru a 

putea crea un argument dinamic în cadrul funcţiei ROW. 

Pe baza aceluiași set de date, calculăm suma primelor 5 vânzări din anul 2014. Iar 

numărul 5 va fi introdus în celula O36. 

{=SUM(LARGE($G$32:$G$43,ROW(INDIRECT("1:"&$O$36))))} 

La fel ca în exemplul anterior, formula este una de tip Array și este nevoie să folosim  la 

final CTRL + SHIFT + ENTER. 

Prin înlocuirea funcţiei LARGE cu SMALL putem afla valoarea ultimelor n valori din listă, 

unde n este un număr introdus de utilizator în altă celulă.  

Elemente de bază ale funcţiilor de tip COUNT 

Sintaxa funcţiei COUNT () 

Funcţia Excel COUNT se folosește pentru a afla câte celule conţin numere din zona de 

celule dorite. Sintaxa aceste funcţii este: 

COUNT(value1,value2,value3, ... ) 

Putem introduce până la 255 de argumente și fiecare dintre ele poate face referinţă la o 

celulă sau o zonă de celule. 

Trebuie reţinut faptul că funcţia COUNT nu va număra celulele care conţin valorile logice 

(True sau False), cele cu text, erori sau cele goale. Deoarece programul Excel stochează 

datele ca numere, celulele care conţin date vor fi numărate dacă folosim funcţia COUNT. 

Exemplu 19 – Numărarea valorilor numerice dintr-o zonă de celule 

Pe baza tabelului 1 de date din sheet-ul COUNT vom calcula numărul de celule numerice 

din tot tabelul. 

=COUNT($C$7:$G$20) 

Sintaxa funcţiei COUNTA () 

Funcţia Excel COUNTA se folosește pentru a afla numărul de celule care conţin orice fel 

de date, inclusiv erori. Sintaxa pentru această funcţie este: 

COUNTA(value1,value2,value3, ... ) 

Putem introduce până la 255 de argumente și fiecare dintre ele poate face referinţă la o 

celulă sau o zonă de celule. 

Exemplu 20 – Numărarea celulelor care conţin date dintr-o zonă de celule 

Folosind același set de date, vom calcula numărul de celule care conţin informaţii din tot 

tabelul. Pentru că nu există celule goale, la rezultatul de la exemplul anterior vom 

adauga 14, reprezentând numărul de celule text de pe coloana Produs. 

=COUNTA($C$7:$G$20) 

Sintaxa funcţiei COUNTBLANK () 

Funcţia COUNTBLANK se folosește pentru a afla numărul de celule goale din zona de 

celule selectată. 

   COUNTBLANK(range) 

După cum vedeţi sintaxa este foarte ușoară. Unicul argument este reprezentat de zona 

de celule unde dorim să aflăm numărul de celule goale. 

Exemplu 21 – Numărarea celulelor goale dintr-o zonă de celule 

Pe baza tabelului de date numărul 2 din sheet-ul COUNT dorim să aflăm câte produse 

mai sunt de livrat.  

=COUNTBLANK($F$27:$F$40) 

 

Sintaxa funcţiei COUNTIF () 

Funcţia COUNTIF se folosește pentru a afla numărul de celule care îndeplinesc o 

condiţie dată de utilizator. 

Sintaxa acestei funcţii este asemănătoare cu cea a funcţiei SUMIF: 

COUNTIF(range, criteria) 

• Range – reprezintă zona de celule care va fi evaluată în funcţie de condiţia notată 

în argumentul Criteria. 

• Criteria – reprezintă condiţia care definește celulele care vor fi contorizate de 

către funcţie. Condiţia dorită poate să fie de tip numeric, text, de tip dată, valoare 

logică sau reprezentând altă celulă. Acţiunea implicită pentru compararea 

celulelor este de tip “egal cu” și nu este nevoie să fie specificată de către utilizator. 

Pentru folosirea operatorilor de comparare ( >,  <,  >=,  <=, <>) trebuie scris între 

ghilimele “>”. 

Exemplu 22 – Numărarea celulelor în funcţie de o celulă 

Pe baza tabelului de date numărul 2 din sheet-ul COUNT, vom pune în aplicare cel mai 

simplu exemplu de folosire a funcţiei COUNTIF. Vom scrie o formulă pentru a afla 

numărul de livrări pentru produsul Capsator. 

=COUNTIF($D$27:$D$40,"Capsator") 

 

Exemple 23-25  Mai mic sau mai mare decât o valoare dată 

În continuare vom trece prin câteva exemple, în care vom vedea cum se folosesc 

operatorii de comparare specificaţi mai sus:  

  



 FUNCŢIILE EXCEL DIN GRUPURILE COUNT ȘI SUM 

 
 26 | P a g e  

Condiţia Operator Formula exemplu Descriere 

Mai mare 

decât o 

valoare 

> =COUNTIF($E$27:$E$40,">100") 

Numără toate valorile de pe 
coloana Cantitate care sunt 
mai mari ca 100. 

Mai mare sau 

egal ca o 

valoare 

>=  Funcţionează la fel ca operatorul > 

Mai mic decât 

o valoare 
< =COUNTIF($E$27:$E$40,"<20") 

Numără toate valorile de pe 
coloana Cantitate care sunt 
mai mici de 20. 

Mai mic sau 

egal ca o 

valoare 

<=  Funcţionează la fel ca operatorul < 

Diferit de o 

valoare 
<> 

=COUNTIF($D$27:$D$40,"<>Pix 

negru") 

Numără toate livrările care 
nu conţin produsul “Pix 
negru” 

 

Exemple 26 Numărarea parţială folosind caracterul tip Wildcard * 

În cazul în care tabelul de date conţine diverse variaţii ale unui cuvânt cheie care ne 

interesează, putem să folosim caracterul * pentru a număra celulele care îl conţin. 

Pe baza tabelului de date numărul 2 din sheet-ul COUNT dorim să aflăm câte livrări cu 

produse care conţin cuvântul adeziva au fost realizate. 

=COUNTIF($D$27:$D$40,"*adeziva*") 

Funcţia COUNTIF verifică fiecare celulă din zona de celulele selectate și va contoriza 

orice text conţine cuvântul adeziva. 

Exemple 27 Numărarea celulelor care încep sau se termină cu anumite caractere * 

Folosind același caracter wildcard * pentru a afla numărul de celule care încep sau se 

termină cu un anumit text dat. Folosind același set de date vom utiliza formula 

următoare pentru a afla numărul de livrări pentru produse care încep cu cuvântul Pix: 

=COUNTIF($D$27:$D$40,"Pix*") 

Sintaxa funcţiei COUNTIFS () 

Funcţia COUNTIFS se folosește pentru a afla numărul de celule care îndeplinesc una sau 

mai multe condiţii. Această funcţie a fost introdusă din versiunea Excel 2007 și are 

sintaxa asemănătoare cu cea a funcţiei COUNTIF: 

COUNTIFS(criteria_range1, criteria1, [criteria_range2, criteria2]…) 

• Criteria_range1 – definește prima zonă de celule pentru care se aplică prima 

condiţie setată de utilizator; 

• Criteria1 – reprezintă condiţia care se aplică pentru zona de celule aleasă 

anterior; 

• [criteria_range2, criteria2]… – reprezintă intervale adiţionale de date și criteriile 

asociate acestora. Se pot specifica până 127 de perechi de criterii în cadrul 

formulei. 

Ce trebuie să știm despre funcţia COUNTIFS: 

• Putem folosi această funcţie atât pentru a aplica o condiţie, cât și pentru mai 

multe condiţii. 

• Fiecare zonă de celulă adiţională aleasă în cadrul formulei trebuie să aibă același 

număr de rânduri ca prima zonă aleasă (argumentul criteria_range1). 

• Dacă unul din criterii face referire la o celulă goală, funcţia o consideră ca fiind 

valoarea zero (0); 

 

Exemplu 28 – Numărarea celulelor care se încadrează într-un interval dat 

Pe baza tabelului de date numărul 1 din sheet-ul COUNT, vom folosi funcţia COUNTIFS 

pentru a afla câte din livrările realizate au avut valori între 100 și 200. 

=COUNTIFS($G$7:$G$20,">100",$G$7:$G$20,"<200") 

Exemplu 29 – Numărarea celulelor care îndeplinesc simultan mai multe condiţii 

Folosind același set de date, vom calcula numărul livrărilor care s-au realizat după 

01.05.2015, inclusiv această dată, având valoarea mai mare de 200, inclusiv această 

valoare. De această dată, nu vom nota criteriile în cadrul formulei, ci ne vom folosi de 

două celule pentru a face acest lucru. 

=COUNTIFS($C$7:$C$20,">=" &$J$16,$G$7:$G$20,">="&$L$16) 

Utilizări avansate ale funcţiilor din grupul 

COUNT 

Numărarea valorilor care se încadrează într-un interval dat 

(COUNTIF) 

Mai sus am văzut cât de mult ne ajută COUNTIFS pentru analizarea datelor dintr-un 

tabel de date. Pentru că nevoia de a număra celulele care îndeplinesc simultan mai 

multe condiţii exista și înainte de introducerea funcţiei COUNTIFS, vom vedea cum 

putem face acest lucru folosindu-ne de mai multe funcţii COUNTIF. 

Exemplu 30 – Numărarea valorilor care se încadrează într-un interval dat folosind 

COUNTIF 

Pentru acest exemplu, vom folosi tabelul de date 1 din sheet-ul COUNT Avansat. În 

cadrul acestui tabel sunt înregistrate cheltuielile realizate într-o  anumită perioadă. Pe 

baza acestor informaţii, noi dorim să aflăm câte cheltuieli s-au realizat de pe 05.01.2015 

până 10.01.2015: 

=COUNTIF($C$8:$C$28,">="&$J$8)-COUNTIF($C$8:$C$28,">"&$L$8) 

Prima funcţie COUNTIF contorizează toate valorile de pe coloana Data care sunt mai 

mari sau egale cu 05.01.2015. (rezultat 18) 

Din rezultatul primit, scădem numărul de valori care sunt mai mari de data de final 

analizată, 10.01.2015. (rezultat 7) 

Numărarea valorilor care îndeplinesc cel puţin una din condiţiile 

setate 

În exemplele anterioare, am văzut cum putem folosi COUNTIF sau COUNTIFS pentru a 

contoriza celulele care îndeplinesc în același timp mai multe condiţii.  

Ne putem întreba cum facem atunci când dorim să aflăm numărul de celule când este 

îndeplinită cel puţin una din condiţiile setate. De exemplu, cum putem calcula numărul 

total al zonelor cu vânzări mai mari de 1500 în anii 2012 – 2014. Pentru acest exemplu 

vom folosi tabelul 2 din sheet-ul COUNT Avansat. 

Pentru a afla acest lucru, vom folosi trei funcţii COUNTIF pentru a testa fiecare coloană 

aferentă fiecărui an. Pentru că ne interesează ca valorile să îndeplinească cel puţin o 

condiţie, vom aduna rezultatele primite.  

=COUNTIF($F$33:$F$44,">1400")+COUNTIF($G$33:$G$44,">1400")+ 

COUNTIF($E$33:$E$44,">1400") 

 

Aflarea numărului de valori care se regăsesc doar o dată într-o 

zonă de celule 

Mai putem folosi funcţia COUNTIF pentru a afla numărul valorilor unice dintr-o zonă de 

celule selectate. Pentru a realiza acest lucru, este nevoie să folosim o coloană ajutătoare. 

Exemplu 32 – Numărarea valorilor care se regăsesc doar o dată într-o zonă de celule 

Pentru acest exemplu, vom folosi tabelul de date 1 din sheet-ul COUNT Avansat și 

dorim să aflăm numărul valorilor unice de pe coloana Categorie.  

Primul pas în acest sens, este ca după coloana Valoare să  inserăm una nouă pe care 

am denumit-o Duplicat. Folosind funcţia COUNTIF pentru a verifica pe fiecare rând 

situaţia în care categoria respectivă se repetă în tabel. Dacă se repetă, rezultatul dat de 

COUNTIF va fi mai mare de 1, iar dacă nu se repetă va fi zero. 

=COUNTIF($E$8:$E$28,E8)>1 

Prin compararea rezultatului dat de funcţia COUNTIF cu valoarea 1, vom primi True 

pentru categoriile care se repetă pe această coloană, și False pentru cele unice. 

Pentru a afla răspunsul la întrebarea noastră, tot ce ne rămâne de făcut este să 

numărăm câte dintre aceste valori sunt False. 

=COUNTIF($H$8:$H$28,"FALSE") 

